Appendix CIDI 2001-2002


NHANES Composite International Diagnostic Interview - 
Panic Disorder Module (CIQPAN_B)
Survey Years Included in this File: 2001-2002
March 2006

The following algorithms were used to determine the panic score variable.


libname nh6 xport 'C:\nhanes\ciqpan_b.xpt';

data ciqpan_b; 
  set nh6.ciqpan_b; 
  if n(of CIQP:)=0 then return;

/* Panic Disorder */

/* A. Intense period of fear of discomfort that is recurrent and occur in
unexpected situation.  Four or more attack symptoms.  At least 1 attack
followed by a month of concern, worry, or behavior change.*/

  cntpdisa=0;
  if CIQP13A=1 
       then cntpdisa=cntpdisa+1;
  if CIQP13B=1 
       then cntpdisa=cntpdisa+1;
  if CIQP13C=1 
       then cntpdisa=cntpdisa+1;
  if CIQP13E=1 
       then cntpdisa=cntpdisa+1;
  if CIQP13F=1 
       then cntpdisa=cntpdisa+1;
  if CIQP13G=1 
       then cntpdisa=cntpdisa+1;
  if CIQP13H=1 
       then cntpdisa=cntpdisa+1;
  if CIQP13I=1 
       then cntpdisa=cntpdisa+1;
  if CIQP13J=1 or CIQP13K=1 
       then cntpdisa=cntpdisa+1;
  if CIQP13L=1 
       then cntpdisa=cntpdisa+1;
  if CIQP13N=1 
       then cntpdisa=cntpdisa+1;
  if CIQP13O=1 
       then cntpdisa=cntpdisa+1;
  if CIQP13P=1 
       then cntpdisa=cntpdisa+1;

  if CIQP03=1 and 1<CIQP14<7777 and cntpdisa>=4 
     and CIQP15 in(1,2) and CIQP12^=2 and 1<CIQP32<7777 
     and (CIQP04=1 or CIQP06=1 or CIQP08=1) 
       then pdisa=1;
       else pdisa=5;


/* C. Panic attacks are not due to a direct physiological effect of a
      substance of general medical condition */

  if CIDPPRB in(3,4) 
       then pdisc=5;
       else pdisc=1;

  if pdisa=1 and pdisc=1 
       then pdis=1;
       else pdis=5;

  cidpscor = pdis;

run; 




NHANES Composite International Diagnostic Interview - CIDI
Generalized Anxiety Disorder Module (CIQGAD_B)
Survey Years Included in this File: 2001-2002
March 2006

The following algorithms were used to determine the generalized anxiety disorder score variable.


libname nh4 xport 'C:\nhanes\ciqgad_b.xpt';

data ciqgad_b; 
  set nh4.ciqgad_b; 
  if n(of CIQG:)=0 then return;


/* Generalized Anxiety Disorder */

/* A. Excessive anxiety and worry */
  if CIQG01=1 and (CIQG02=1 or 6<=CIQG03<77)
     and CIQG04 in(1,2,3) 
       then gada=1;
   else if CIQG06=1 and (CIQG07=1 or 6<=CIQG08<77) 
     and CIQG09 in(1,2,3) 
       then gada=1;
       else gada=5;

  if CIQG13=1 then gada=5;

/* B. Difficult to control the worry */
  if CIQG15 in(1,2) or CIQG16 in(1,2) 
       then gadb=1;
       else gadb=5;

/* C. Anxiety or worry is associated with 3/6 symptoms */
  cntgadc=0;
  if CIQG17A=1 or CIQG17B=1 
       then cntgadc=cntgadc+1;
  if CIQG17C=1 
       then cntgadc=cntgadc+1;
  if CIQG17F=1 
       then cntgadc=cntgadc+1;
  if CIQG17D=1 
       then cntgadc=cntgadc+1;
  if CIQG17G=1 
       then cntgadc=cntgadc+1;
  if CIQG17E=1 
       then cntgadc=cntgadc+1;
  if cntgadc>=3 
       then gadc=1;
       else gadc=5;

/* D. The worry is not exclusively about a mental disorder, weight gain,
      or multiple physical complaints. */

  if CIQG12=1 
       then gadd=5;
       else gadd=1;

/* E. The anxiety causes clinically significant distress or impairment */
  cntgade=0;
  if CIQG26 not in(777,999,.) 
       then cntgade=cntgade+CIQG26;
  if CIQG29 not in(777,999,.) 
       then cntgade=cntgade+CIQG29;
  if CIQG33 not in(777,999,.) 
       then cntgade=cntgade+CIQG33;
  if CIQG36 not in(777,999,.) 
       then cntgade=cntgade+CIQG36;

  if CIQG24=1 or CIQG25=1 or cntgade >=6 
       then gade=1;
       else gade=5;

/* F. The disturbance is not due to the direct physiological effects of a
      substance or general medical condition */
  if CIDGPRB in(3,4) 
       then gadf=5;
       else gadf=1;

  if gada=1 and gadb=1 and gadc=1 
     and gadd=1 and gade=1 and gadf=1
       then gad=1;
       else gad=5;

  cidgscor = gad;

run;




NHANES Composite International Diagnostic Interview - CIDI
Major Depression Module (CIQDEP_B)
Survey Years Included in this File: 2001-2002
March 2006

The following algorithms were used to determine the major depression score variable.


libname nh5 xport 'C:\nhanes\ciqdep_b.xpt';

data ciqdep_b; 
  set nh5.ciqdep_b; 
  if n(of CIQD:)=0 then return;

/* Major Depression*/

/* A. Five (or more) of the following symptoms have been present
   during the same 2-week period and represent a change from previous
   functioning; at least one of the symptoms is either:
*/

  depacnt=0;
/* (1) Depressed mood */
  if (CIQD001=1 and CIQD002 in(1,2) and CIQD003 in(1,2)) or
     (CIQD014=1 and CIQD015 in(1,2) and CIQD016 in(1,2)) 
       then depacnt=depacnt+1;

/* (2) Markedly diminished interest */
  if CIQD006=1 or (CIQD008=1 and CIQD009 in(1,2) 
     and CIQD010 in(1,2)) or CIQD031=1
       then depacnt=depacnt+1;

  if depacnt>0 then do;

  /* (3) significant appetite/weight change */
    if CIQD019=1 or 10<=CIQD021<777 or CIQD022=1 
       or 10<=CIQD024<777 
         then depacnt=depacnt+1;

  /* (4) Insomnia or Hypersomnia */
    if CIQD026 in(1,2) or CIQD027=1 or CIQD028=1 
         then depacnt=depacnt+1;

  /* (5) Psychomotor agitation or retardation */
     if CIQD033=1 or CIQD035=1 
         then depacnt=depacnt+1;

  /* (6) Fatigue or loss of energy */
     if CIQD005=1 or CIQD012=1 or CIQD018=1 
         then depacnt=depacnt+1;

  /* (7) Worthlessness of quilt */
     if (CIQD036=1 or CIQD037=1 or CIQD041=1) 
        and CIQD040^=1 
          then depacnt=depacnt+1;

  /* (8) Diminished ability to think or concentrate or indecisiveness */
     if CIQD042=1 or CIQD043=1 or CIQD044=1 
        or CIQD045=1 or CIQD046=1 
          then depacnt=depacnt+1;

   /* (9) Recurrent thoughts of death, suicidal ideation, plan or attempt */
      if CIQD047=1 or CIQD048=1 or CIQD049=1 or CIQD050=1 
          then depacnt=depacnt+1;
  end;
  if depacnt >=5 
       then depa=1;
       else depa=5;


/* C. The symptoms cause clinically significant distress or
   impairment.
*/

  cntdepc=0;
  if CIQD096 not in(777,999,.) 
      then cntdepc=cntdepc+CIQD096;
  if CIQD099 not in(777,999,.) 
      then cntdepc=cntdepc+CIQD099;
  if CIQD103 not in(777,999,.) 
      then cntdepc=cntdepc+CIQD103;
  if CIQD106 not in(777,999,.) 
      then cntdepc=cntdepc+CIQD106;

  if CIQD095=1 or CIQD095=2 or cntdepc>=6 
      then depc=1;
      else depc=5;

/* D. The symptoms are not due to the direct physiological effects of a
   substance or general medical condition.
*/
  if CIDDPRB in(3 4) 
     then depd=5;
     else depd=1;

/* E. Symptoms are not better accounted for by bereavement.
      Bereavement depression does not count if it lasts 2 months or less
      or is not characterized by marked impairment, suicidal ideation,
      or psychomotor retardation.

      If respondent answers "don't know" to a bereavement question we are
      assuming the period of depression in question did not begin just after
      someone close died.  If respondent refuses to answer a bereavement
      question, we are assuming the period of depression in question did begin
      just after someone close died.
*/
  if depc=1 or CIQD048=1 or CIQD033=1 
     or CIQD049=1 or CIQD050=1 or CIQD036=1 
       then overide=1; 
       else overide=5;

/* Single episode */
  if CIQD052=1 or CIQD053=1 then do;
    if CIQD059 in(2,9) 
        then depe=1;
       else if CIQD059 in(1,7) and (8 < CIQD051 <=52 or overide=1)
        then depe=1;
       else depe=5;
  end;

/* Two episodes */
  else if (CIQD052=2 or CIQD053=2) and CIQD064=2 then do;
     if CIQD066 in(2,9) or CIQD078 in(2,9) 
        then depe=1;
     else if (CIQD065*7>56 or CIQD075>56) or overide=1 
        then depe=1;
     else depe=5;
  end;

/* More than 2 episodes */
  else if (CIQD052 in(7,9) or CIQD053 in(7,9)) or
      (CIQD052=2 or CIQD053=2 and CIQD064 >=3) then do;
    if CIQD091 in(2,9) then depe=1;
       else if CIQD083*7>60.8 or overide=1 then depe=1;
       else depe=5;
  end;
  else depe=5;


/* Major Depression Diagnosis */
  if depa=1 and depc=1 and depd=1 and depe=1 
     then dep=1;
     else dep=5;
  
  ciddscor=dep;

run;